

# 物料在振动运输机上的运动过程

机械系 李松涛

## 提 要

本文分析了载运过程中的滑动现象,导出考虑滑动时计算平均输送速度的方法,并编制了计算最佳工作状态下的斜置角,输送速度、动力消耗系数以及其他运动学参数的 BASIC 程序。

## 前 言

从六十年代初以来,对于振动运输机的输送过程已经进行了许多研究。以往的大多数研究中都假定在载运过程(即物料和槽体接触的阶段)无相对运动,因而被称为“粘着阶段”。少数研究考虑了载运过程存在滑动现象。但是考虑载运过程的滑动现象来确定物料起跳时的水平速度,首先必须确定各滑动阶段的起迄时间,起迄时间的确定必须进行非常繁复的计算。由于计算的困难以往的研究都假定物料起跳时的水平速度与槽体相等。实际上振动运输机上的物料,在载运过程不可能没有滑动。既有滑动存在,物料起跳时的水平速度就不会与槽体相等。

本文着重分析了载运过程中的滑动现象,应用电子计算机计算了各滑动阶段的起迄相位角,导出考虑滑动时计算平均输送速度的方法,并编制了根据机器指数和物料与槽体之间的摩擦系数来计算最佳工作状态下的斜置角,平均输送速度,动力消耗系数以及其他运动学参数的 BASIC 程序,以供设计计算振动运输机时应用。

---

(接前页)

## 参 考 文 献

- [1]形状和位置公差 术语及定义(GB1183—80)
- [2]形状和位置公差 检验规定 (GB1958—80)
- [3]公差与配合(GB1800—1804—79)
- [4]极限量规
- [5]Machine Design Vol.47,48.
- [6]相关公差的量规设计 张瑞林:《形位误差测试技术综合参考资料》1977年

## 一、理论输送速度

图 1 为振动输送机简图。当该振动输送机沿 S 方向作简谐运动时, 物料 A 在一定运动参数的条件下, 便作周期性的跳跃, 同时沿 X 方向向前传送。若槽体的运动方程为:

$$S = A \sin \omega t$$




图 1

则槽体和物料在垂直方向的位移关系如图 2 所示。图中  $p(p_1, p_2)$  为物料开始跳跃时槽体所处的相位角, 称跳始角。  $z(z_1, z_2)$  为跳跃终止时槽体所处的相位角, 称跳迄角。设  $K$  为机器指数, 表示槽体加速度最大值与重力加速度之比, 即

$$K = \frac{A \omega^2}{g}$$

设  $\Gamma$  为抛射指数, 表示槽体垂直方向的加速度与重力加速度之比, 即

$$\Gamma = K \sin \alpha$$

据此可以导出以下的关系: [5]

$$\operatorname{ctg} p = \frac{\frac{\theta^2}{2} - (1 - \cos \theta)}{\theta - \sin \theta} \quad \dots\dots\dots (1)$$

$$\sin p = \frac{1}{\Gamma} \quad \dots\dots\dots (2)$$

其中  $\theta = z - p$

通过计算可知:  $\Gamma = 1$  时  $\theta = 0$ ,  
 $\Gamma = 3.3$  时  $\theta = 2\pi$

所以物料在每个周期发生一次跳跃的条件为抛射指数在 1—3.3 范围之内, 这是大多数振动输送机的取值范围。

设计振动输送机要进行物料输送速度的计算。物料的输送速度通常是根椐理论输送速度修正而得, 理论输送速度常以下列两公式计算

$$1. \quad V_{th} = \frac{gn^2}{2f} \operatorname{ctg} \alpha. \quad [1]$$

$$\text{或} 2. \quad V_{th} = A \omega \cos \alpha \cos \beta \quad [4]$$

其中  $f = \frac{\omega}{2\pi}$

$g$ ——重力加速度

$n$ ——为物料跳跃过程所占时间与振动周期之比

$\alpha$ ——斜置角(见图 1)

以上两公式都是简化了运动过程而导出的。其共同特点是把物料起跳时水平速度假定为与槽体相等。这对振动输送机一般是不可能的。它们的不同点是前者：假定物料落到槽体上以后即粘附于槽体，无相对滑动。也就是相当于摩擦力极大的情况。后者：假定落在槽体上的物料仍以不变的水平速度向前运动。即相当于摩擦系数为零的情况。

为了进一步研究物料的理论输送速度，考虑滑动影响，下面将对载运过程进行研究分析。设如图 1 所示的振动输送机。其运动方程式为：

$$y = A \sin \alpha \sin \omega t \dots \dots \dots (3)$$

$$X = A \cos \alpha \sin \omega t \dots \dots \dots (4)$$

若某质量为  $m$  的物料颗粒，由于摩擦力的作用粘附在槽体上，它将受到交变的惯性力的作用。惯性力  $Q$  为

$$Q = mA\omega^2 \sin \omega t$$

$Q$  在  $X$  方向的分量为

$$Q_x = mA\omega^2 \cos \alpha \sin \omega t, \quad \text{令 } \omega t = \phi$$

当  $\phi = \frac{\pi}{2}, \frac{3\pi}{2}, \frac{5}{2}\pi$  等值时  $Q_x$  的绝对值最大。故在以上诸点附近的区域中，

(见图 3 中  $b-b_1$  和  $d-d_1$ )  $Q_x$  足以克服物料与槽体之间的摩擦力而使物料发生滑动。槽体加速度为正时发生反滑，即向  $X$  的负方向滑动。槽体加速度为负时发生正滑，即向  $X$  的正向滑动。

反滑开始时惯性力与摩擦力最大值相等，方向相反。根据动力学的平衡原理存在如下关系：




图 2




图 3

$$\mu (mg - mA\omega^2 \sin \alpha \sin b) = -mA\omega^2 \cos \alpha \sin b$$

$$\sin b = \frac{-\mu \tan \alpha}{1 - \mu \tan \alpha}$$

正滑开始时存在:

$$\mu (mg - mA\omega^2 \sin\alpha \sin d) = mA\omega^2 \cos\alpha \sin d$$

$$\sin d = \frac{\mu \tan \alpha}{\Gamma (1 + \mu \tan \alpha)} .$$

其中  $\mu = \tan \rho$ ,  $\mu$  为摩擦系数,  $\rho$  为摩擦角

$$\text{令 } \tan \rho \cdot \tan \alpha = R$$

$$\text{则 } \sin b = \frac{-R}{\Gamma (1 - R)} ,$$

$$\sin d = \frac{R}{\Gamma (1 + R)} .$$

显然:

$$b_1 = 3\pi - b$$

$$d_1 = 3\pi - d$$

$b, b_1$  各为静态反滑始角和静态反滑迄角;  $d, d_1$  各为静态正滑始角和静态正滑迄角。区域  $b-b_1$  为静态反滑区,  $d-d_1$  为静态正滑区。

在振动运输机运行中, 跳跃中的物料颗粒落下来与槽体接触时, 其水平即时速度一般并不与槽体相等, 所以即使不在静态滑动区也会发生滑动。而且滑动一旦发生, 也不可能在  $b_1$  或  $d_1$  时遽然停止。所以实际的, 即动态的滑动区并不与静态滑动区重合。

物料从落到槽体上以后到再次被抛起以前那段时间内的过程称为载运过程。振动运输机输送物料时抛掷过程与载运过程交替进行。在载运过程中物料于不同的时间可能发生正滑和反滑两种不同的滑动。两种滑动互相过渡时一定要经过物料与槽体速度相同的点, 也就是一种滑动的终止点。在滑动的终止点物料与槽体的速度相等, 是相对静止的。相对静止能够保持的条件是滑动终止点落在静态滑动区以外。否则将立即发生另一种方向的滑动。

设有一被抛掷而处于跳跃状态的物料颗粒, 其水平速度为  $V_F$ , 在  $\phi = z$  时落于槽底, 且  $V_F > \dot{X}_z$ 。 ( $\dot{X}_z = A\omega \cos\alpha \cos z$ , 为  $\phi = z$  时槽体的即时速度。) 这时物料便发生正滑, 由于存在摩擦, 正滑的速将逐渐减小, 至  $\phi = a$  时为零, 则  $a$  为正滑迄角。这时:

$$V_a = \dot{X}_a$$

$V_a$ ——正滑终止点的物料水平速度

$\dot{X}_a$ ——正滑终止点槽体的水平速度

由  $\phi = z$  到  $\phi = a$  为初正滑阶段。若  $a < b$ , 则在  $a$  到  $b$  之间是能保持相对静止的附着阶段。若  $a \geq b$ , 则  $a$  点落在静态反滑区内便不存在附着阶段。随后物料进入反滑阶段。令反滑阶段起点的相位角为  $u$ , 则

$$u = \begin{cases} b & \text{若 } a < b \\ a & \text{若 } a \geq b, \end{cases}$$

反滑经过一个由相对速度为零到最大值而又回到零的过程。若  $\phi = c$  时反滑终止，则  $c$  为反滑迄角。这时：

$$V_c = \dot{X}_c$$

$V_c$ ——反滑终止时物料的水平速度

$\dot{X}_c$ ——反滑终止时槽体的水平速度。

由  $\phi = u$  到  $\phi = c$  为反滑阶段。若  $c < d$  则  $c$  到  $d$  之间为附着阶段，若  $c \geq d$  则无此附着阶段。随后是终正滑阶段。终正滑延续到物料再次起跳为止。这时  $\phi = 2\pi + p$ 。令终正滑的起点为  $v$ ，则

$$v = \begin{cases} d & \text{若 } c \leq d \\ c & \text{若 } c > d \end{cases}$$

以上所述为一典型的过程。存在初正滑、附着、反滑、附着、终正滑 5 个阶段。运动学参数各异的振动运输机，并非都存在全部五个阶段，有的只存在三个、甚至两个阶段。

物料在正滑时的运动方程为：

$$m \frac{dV_{\text{正}}}{dt} = -m\mu (g - A\omega^2 \sin\alpha \sin\omega t)$$

$$dV_{\text{正}} = -\mu (g - A\omega^2 \sin\alpha \sin\omega t) dt$$

$$= -\frac{\mu}{\omega} (g - A\omega^2 \sin\alpha \sin\phi) d\phi$$

$$V_{\text{正}} = -\frac{\mu g}{\omega} \phi - \mu A \omega \sin\alpha \cos\phi + c \dots \dots \dots (7)$$

根据物料在反滑时的运动方程同样可得：

$$V_{\text{反}} = \frac{\mu g}{\omega} \phi + \mu A \omega \sin\alpha \cos\phi + c \dots \dots \dots (8)$$

设初正滑阶段物料的水平速度为  $V_1$ ，初始条件为：  $\phi = z, V_1 = V_p$

令  $V_p = k\dot{X}_p$ ，为物料跳跃状态下的水平速度。 $\dot{X}_p$  为物料起跳时槽体的水平速度。

$$\dot{X}_p = A\omega \cos\alpha \cos p$$

根据式 (7) 可得：

$$V_1 = A\omega (k \cos p + \frac{R}{l} z + R \cos z - \frac{R}{l} \phi - R \cos \phi) \cos\alpha \dots \dots \dots (9)$$

又根据定义,  $\phi = a$ 时,  $V_1 = \dot{X}_a = A\omega \cos \alpha \cos a$  代入式(9)得:

$$k \cos p + R \cos z + \frac{R}{\Gamma} z - (1 + R) \cos a - \frac{R}{\Gamma} a = 0 \dots\dots\dots (10)$$

设反滑阶段物料的水平速度为 $V_2$ , 初始条件为:  $\phi = u, V_2 = \dot{X}_u$

$\dot{X}_u$ 为反滑起点槽体的水平速度

$$\dot{X}_u = A\omega \cos \alpha \cos u$$

根据式(8)可得:

$$V_2 = A\omega \cos \alpha \left[ (1 - R) \cos u - \frac{R}{\Gamma} u + R \cos \phi + \frac{R}{\Gamma} \phi \right] \dots\dots\dots (11)$$

又根据定义  $\phi = c$ 时,  $V_2 = \dot{X}_c = A\omega \cos \alpha \cos c$ 。代入(11)得:

$$(1 - R) \cos u - \frac{R}{\Gamma} u - (1 - R) \cos c + \frac{R}{\Gamma} c = 0 \dots\dots\dots (12)$$

设终正滑阶段的物料水平速度为 $V_3$ , 初始条件为:  $\phi = v, V_3 = \dot{X}_v$

$\dot{X}_v$ 为终正滑起点槽体的水平速度。

$$\dot{X}_v = A\omega \cos \alpha \cos v$$

根据式(7)可得:

$$V_3 = A\omega \cos \alpha \left[ (1 + R) \cos v + \frac{R}{\Gamma} v - R \cos \phi - \frac{R}{\Gamma} \phi \right] \dots\dots\dots (13)$$

设再次起跳时物料的水平速度为 $V'_p$ , 令 $V'_p = k' A\omega \cos c \cos p$ , 当  $\phi = 2\pi + p$ 时,

则 $V_3 = V'_p$

根据式(13)可得:

$$k' = \frac{1}{\cos p} \left[ (1 + R) \cos v - \frac{R}{\Gamma} (2\pi + p - v) \right] - R \dots\dots\dots (14)$$

根据抛射指数 $\Gamma$ 与跳始角 $p$ 和跳迄角 $z$ 的关系[见公式(1)和公式(2)], 当 $\Gamma$ 为2.98时 $z = 360^\circ$ , 槽体的水平速度达到最大值即 $\dot{X}_{360^\circ} = A\omega \cos \alpha$ 。在这一点附近物料水平速度将小于槽体的水平速度而发生初反滑阶段。初反滑阶段的初始条件为:  $\phi = z$ ,

$V_2 = V_p$ 。

根据式(8)可得:

$$V_2 = A\omega \cos \alpha \left[ k \cos p - R \cos z - \frac{R}{\Gamma} z - (1 - R) \cos \phi + \frac{R}{\Gamma} \phi \right] \dots\dots\dots (15)$$




图 4 程序框图

又当  $\phi = c$  时,  $V_2 = Xc$ , 代入 (15) 得:

$$K \cos p - R \cos z - \frac{R}{\Gamma} z - (1 - R) \cos c + \frac{R}{\Gamma} c = 0 \dots (16)$$

当  $R$  值和  $\Gamma$  值都很小时, 由于滑动的趋势和相对速度都增加了, 所以扩大了反滑区, 使反滑区一直延续到再次起跳。

这时  $\phi = 2\pi + p$ ,  $V_2 = K' A \omega \cos \alpha \cos p$ , 代入 (11) 得:

$$k' = \frac{1}{\cos p} \left[ (1 - R) \cos u + \frac{R}{\Gamma} (2\pi + p - u) \right] + R \dots (17)$$

根据以上推导, 若假定一个  $k$  值, 就可以求得  $k'$ 。  $k$  和  $k'$  都不是稳定输送状态下的精确值。  $k$  的精确值可以用迭代法求得。即求得  $k'$  之后令  $k = k'$ , 再反复计算  $k'$ , 直至  $k' = k$  为止。从物理意义上分析, 这一迭代格式是收敛的, 所以求得精确的  $k$  值是可能的。当然在工程计算中, 只要求精度足够的近似值就可以了。稳定输送状态下的  $k$  值求得以后,  $a$ ,  $c$  等相角也随之确定了。图 4 为这一计算过程的程序框图。图 5 所示为不同  $R$  值时  $Z$ ,  $p$ ,  $a$ ,  $b$ ,  $c$ ,  $d$  等相角与  $r$  的关系, 是用电子计算机计算的部分结果。

一个周期内物料输送的距离, 可以用物料在槽体上的相对位移来表示。因为槽体作周期性往复运动, 一个周期以后又回到原来的位置, 所以物料的相对位移与其绝对位移相等。设初正滑距离为  $l_1$ , 反滑距离为  $l_2$ , 终正滑距离为  $l_3$ , 跳跃距离为  $l_4$ , 一周期内输送的距离为  $l$ 。




图 5 a






图 5 b




图 5 c




图 5 d

则:  $l = l_1 + l_2 + l_3 + l_4$

$$= \int_{z/\omega}^{a/\omega} (V_1 - \dot{X}) dt + \int_{v/\omega}^{c/\omega} (V_2 - \dot{X}) dt + \int_{v/\omega}^{(2\pi+p)/\omega} (V_3 - \dot{X}) dt \\ + kA \cos \alpha \cos p \cdot (z - p) - A \cos \alpha (\sin z - \sin p)$$

$V_1, V_2, V_3$  分别以公式 (9)、(11)、(13) 代入进行积分并化简得:

$$l = A \times M \cos \alpha \dots \dots \dots (18)$$

其中  $M = k(a-p) \cos p + R(a-z) \cos z + (1-R)(c-u) \cos u$

$$+ (1+R)(2\pi+p-v) \cos v + R(\sin z - \sin p) \\ + (1+R)(\sin v - \sin a) - (1-R)(\sin c - \sin u) \\ - \frac{R}{2\Gamma} [a-z)^2 - (c-u)^2 + (2\pi+p-v)^2] \dots \dots \dots (19)$$

物料的平均输送速度  $V_{\text{平}}$  为:

$$V_{\text{平}} = \frac{l}{T} = \frac{\omega}{2\pi} l \dots \dots \dots (20)$$

设  $L_0$  为每振动一次物料输送的距离与振动机振幅的比值称为输送速度系数。

$$L_0 = \frac{l}{A} = \frac{A \times M \cos \alpha}{A} = M \cos \alpha \dots \dots \dots (21)$$

$L_0$  的值取决于  $M$ 。公式 (19) 是由原始数据  $K, \rho, \alpha$  推导来的, 所以  $L_0$  最后取决于机器指数  $K$ 、摩擦角  $\rho$  和斜置角  $\alpha$ 。当  $K$  和  $\rho$  已定时改变  $\alpha$  可得到不同的  $L_0$  值, 与  $L_0$  的最大值对应的  $\alpha$  值为最佳工作点。当  $\rho = 30^\circ$  时, 最佳工作点的  $K, \alpha$  的关系如图 7。

以  $l = AL_0$  代入 (20) 得:

$$V_{\text{平}} = A\omega \frac{L_0}{2\pi} \dots\dots\dots (22)$$

输送速度系数  $L_0$  为  $V_{\text{平}} / A\omega$  的  $2\pi$  倍。

$\frac{L_0}{2\pi}$  与  $K, \alpha$  的关系曲线如图 6




图 6

## 二、振动运输机槽体对物料所作的功和动力消耗系数

振动运输机输送物料时, 槽体要对物料做功。槽体所作之功消耗于为维持一定的输送速度而发生的摩擦和碰撞之中。设一个振动周期所作的功为  $\overline{W}$ , 则

$$\overline{W} = \overline{W}_x + \overline{W}_y$$

$$\overline{W}_x = \begin{cases} (2\pi + p) / \omega \\ f_x(\phi) \dot{x} dt \dots\dots\dots (23) \\ p / \omega \end{cases}$$

$$\dot{x} = A\omega \cos \alpha \cos \phi$$

$f_x(\phi)$  为一段函数, 不同  $\phi$  取值范围有不同的表达式, 是  $x$  方向槽体对物料的作用力。




图 7

$$f_x(\varphi) = \begin{cases} 0, & p < \varphi < z \\ -\mu m(g - A\omega^2 \sin \alpha \sin \varphi), & z \leq \varphi < a \\ -mA\omega^2 \cos \alpha \sin \varphi, & a \leq \varphi \leq u \\ \mu m(g - A\omega^2 \sin \alpha \sin \varphi), & u < \varphi < c \\ -mA\omega^2 \cos \alpha \sin \varphi, & c \leq \varphi \leq v \\ -\mu m(g - A\omega^2 \sin \alpha \sin \varphi), & v < \varphi \leq 2\pi + p \end{cases}$$

对式 (23) 右边积分化简得:

$$\overline{W}_x = A m g F \cos \alpha \cdot \operatorname{ctg} \alpha \cdots \cdots \cdots (24)$$

其中  $F = \Gamma \{ \sin^2 a - \sin^2 u + \sin^2 c - \sin^2 v - R[(\sin p - \sin z)^2 - (\sin p -$

$$\sin a)^2 + (\sin p - \sin c)^2 - (\sin p - \sin u)^2 + (\sin p - \sin v)^2] \} \quad (25)$$

若每振动一次在水平方向功的消耗与振幅和物料重量的乘积之比为  $X_0$ , 则:

$$X_0 = F \cos \alpha \cdot \operatorname{ctg} \alpha \cdots \cdots \cdots (26)$$

$$\overline{W}_y = \int_{p/\omega}^{(2\pi+p)/\omega} f_y(\varphi) \dot{y} dt \cdots \cdots \cdots (27)$$

$$\dot{y} = A \omega \sin \alpha \cos \omega t$$

$f_y(\varphi)$  为  $y$  方向槽体对物料的作用力, 也是分段函数:

$$f_y(\varphi) = \begin{cases} 0 & p \leq \varphi < z \\ m g (1 - \Gamma \sin \varphi) & z < \varphi \leq 2\pi + p \\ m \omega \frac{dV_y}{d\varphi} & \varphi = z \end{cases}$$

$V_y$ ——物料在垂直方向的速度。

对 (27) 积分化简得

$$\overline{W}_y = A mg H \sin \alpha \dots\dots\dots (28)$$

若每振动一次在垂直方向功的消耗与振幅和物料重量的乘积之比为  $Y_0$ ，则，

$$Y_0 = H \sin \alpha \dots\dots\dots (29)$$

$$\text{其中 } H = \Gamma \left[ (\cos z - \cos p + \frac{z+p}{\Gamma}) \cos z + \frac{(\sin p - \sin z)^2}{2} \right] \dots\dots\dots (30)$$

$H$  与  $\Gamma$  的关系曲线如图 8，对于输送易破碎的物料应取较小的  $H$  值。

动力消耗系数定义为每单位重量物料一个周期内所消耗的能量比输送的距离。即

$$K_E = \frac{A_0}{1} \left[ \frac{\text{公斤} \cdot \text{米} (\text{能量})}{\text{公斤} \cdot \text{米} (\text{运输量})} \right] = 2.722 \times 10^{-3} \frac{A_0}{1} \left[ \frac{\text{吨} \cdot \text{小时}}{\text{吨} \cdot \text{米}} \right] \dots\dots\dots (31)$$

$K_E$  也是单位运输量（单位重量的物料输送单位距离）所消耗的能量。




图 8

式中  $A_0$  为单位重量物料一周内所消耗的能量

$$A_0 = \frac{\overline{W}}{mg} = \frac{\overline{W}_x + \overline{W}_y}{mg} = A H \sin \alpha + A F \cos \alpha \tan \alpha \dots\dots\dots (32)$$

代入 (31) 得，

$$K_E = \frac{H \tan \alpha + F \tan \alpha}{M} = \frac{X_0 + Y_0}{L_0} \dots\dots\dots (33)$$

$K_E$ 与 $\Gamma$ ,  $\alpha$ 的关系曲线如图 9

(33) 式计算的动力消耗系数不包括机械传动中的动力消耗。所以比实际的动力消耗系数要小。但可以作为设计时比较方案的参考。现代振动运输机的动力消耗系数为 0.002—0.01 [6]。

## 小 结

在设计振动运输机时首先决定机器指数 $K$ 。机器指数决定惯性力的大小, 因受结构强度的限制不能取得太大。通常取 4—6 [1]。然后以机器指数 $K$ 和物料的摩擦角 $p$ 为原始数据作《振动运输机最佳工作点运动学参数计算程序》的输入变量, 计算最佳工作点的 $\alpha$ ,  $\Gamma$ ,  $K_E L_c$ , 等运动学参数。(见附录)。振幅 $A$ , 圆频率 $\omega$ 的决定要考虑振动对周围设备, 建筑物的影响, 以及振动机的结构方案, 减振方法等因素来选定。平均输送速度根据电子计算机的输出参数 $L_0$ 以及选定的 $A$ ,  $\omega$ 值, 用公式 (22) 计算。将所得的理论值考虑工作面的斜度, 料层厚度等因素加以修正。[5]




图 9

## 附 录

振动运输机最佳工作点运动学参数计算程序:

### 一、应用说明:

本程序应用DJS—130机BASIC语言编写。以机器指数和摩擦系数为原始数据作为程序的输入变量, 经电子计算机运行后可以输出一组最佳工作点的运动学参数。其中包括: 斜置角 $\alpha$ 抛射指数 $\Gamma$ 、速度系数 $L_0$ , 动力消耗系数 $K_E$ 等参数, (详见输出变量表) 供设计计算和研究振动运输机性能时应用。

### 二、输入输出变量表

#### 1. 输入变量表

JO——机器指数 $K$

M9——物料与槽体的摩擦角 $\rho$

E9——对斜置角 $\alpha$ 的精度要求。

## 2. 输出变量表

JK——机器指数 (即JO)

RN——摩擦角 $\rho$  (即M9) [ $^{\circ}$ ]

AL——斜置角 $\alpha$

LO——速度系数 =  $\frac{\text{物料每周期的位移}}{\text{振动运输机的振幅。}}$

KE——动力消耗系数  $\left[ \frac{\text{Kg} \cdot \text{M (能量)}}{\text{Kg} \cdot \text{M (运输量)}} \right]$

$$R = \text{tg} \alpha \times \text{tg} \rho$$

$A_0$ ——每公斤物料每周期消耗的功/振幅

Y—— $\frac{\text{每公斤物料每周期垂直方向消耗的功}}{\text{振动运输机振幅}}$

P——跳始角[ $^{\circ}$ ]

Z——跳迄角[ $^{\circ}$ ]

A——初正滑迄角[ $^{\circ}$ ]

C——反滑迄角[ $^{\circ}$ ]

B——静态反滑始角[ $^{\circ}$ ]

D——静态正滑始角[ $^{\circ}$ ]

N——循环次数

$\bar{W}$ ——类型

G——抛射指数 $\Gamma$

## 三、程序清单

```
0010 INPUT JO,M9,E9
```

```
0012 DEF FNS(X) = ATN(X/SQR(1 - X * X))
```

```
0015 LET A4 = FNS(1.001/JO)
```

```
0016 IF JO < 3.3 THEN 1490
```

```
0018 LET A5 = FNS(3.2969/JO)
```

```
0020 LET Y = 3.14159
```

```
0022 LET M8 = M9 * Y / 180
```

```
0025 LET NO = 1
```

```
0030 LET A6 = A4 + .382 * (A5 - A4)
```

```
0032 LET A8 = A6
```

```
0034 GOSUB 0260
```

```
0036 LET L6 = L0
```

```
0038 LET A7 = A4 + .618 * (A5 - A4)
```

```

0040 LET A8 = A7
0042 GOSUB 0260
0044 LET L7 = L0
0046 IF L6 > L7 THEN 0062
0048 LET A4 = A6
0050 LET A6 = A7
0052 LET L6 = L7
0054 IF A5 - A4 > E9 THEN 0038
0056 LET NO = O
0058 LET A8 = (A5 + A4) ÷ 2
0060 GOSUB 0260
0061 GOTO 1999
0062 LET A5 = A7
0065 IF A5 - A4 ≤ E9 THEN 0056
0070 LET A7 = A6
0075 LET L7 = L6
0080 LET A6 = A4 + .382 * (A5 - A4)
0085 LET A8 = A6
0090 GOSUB 0260
0095 LET L6 = L0
0100 GOTO 0046
0120 LET V2 = COS(V)
0130 LET V1 = SIN(V)
0140 LET U2 = COS(U)
0150 LET U1 = SIN(U)
0160 LET C2 = COS(C)
0170 LET C1 = SIN(C)
0180 LET A2 = COS(A)
0190 LET A1 = SIN(A)
0200 RETURN
0210 LET Z2 = COS(Z)
0220 LET Z1 = SIN(Z)
0230 LET P2 = COS(P)
0240 LET P1 = SIN(P)
0250 RETURN
0260 LET G = JO * SN(A8)
0265 LET P = 1/G
0266 LET P = ATN(P/SQR(1 - P * P))

```



```

0270 LET R = TAN(A8) * TAN(M8)
0275 DEF FNB(X) = ATN((X - SIN(X)) / (X * X/2 + COS(X) - 1)) - P
0280 LET M = Y/2 - P + .1
0281 LET O = 2 * Y
0285 GOSUB 1100
0288 IF J = 8 THEN 1999
0290 LET S = J
0295 LET K = 1
0300 LET Z = P + S
0310 GOSUB 0210
0320 LET N = 0
0330 LET HO = (Z2 - P2 + (Z - P)/G) * Z2 + (P1 - Z1) * (P1 - Z1)/2
0340 LET YO = G * HO * SIN(A8)
0350 DEF FNA(X) = ATN(X/SQR(1 - X * X))
0360 LET X = -R/G/(1 - R)
0370 IF ABS(X) > 1 THEN 1380
0380 LET B = Y - FNA(X)
0390 LET B1 = B * 180/Y
0400 LET X = R/G/(1 + R)
0410 LET D = FNA(X) + 2 * Y
0420 LET D1 = D * 180/Y
0430 IF Z2 > K * P2 THEN 1210
0440 LET W = 0
0450 LET Q = 2 * Y
0460 LET E = -R/G
0470 LET M = Z
0480 LET F = -(1 + R)
0490 LET H = K * P2 + R * Z2 - E * Z
0500 GOSUB 1080
0510 LET A = J
0520 IF A > B THEN 0550
0530 LET U = B
0540 GOTO 0560
0550 LET U = A
0560 IF B = 3 * Y/2 THEN 1030
0570 IF A > 3 * Y - B THEN 1030
0580 LET M = 3 * Y - B
0590 LET Q = 2 * Y + P

```

```

0600 LET E=R/G
0610 LET F=- (1-R)
0620 LET H=- F * CQS(U) - E * U
0630 GCSUB 1080
0640 IF J=8 THEN 1330
0650 LET C=J
0660 IF C>=D THEN 0690
0670 LET V=D
0680 GOTO 0700
0690 LET V=C
0700 LET L=((1+R) * COS(V) - R * (2 * Y+P-V)/G)/P2-R
0710 LET N=N+1
0720 IF N=35 THEN 0770
0730 LET I=K-L
0740 IF ABS(I)<=.0001 THEN 0770
0750 LET K=L
0760 GOTO 0430
0770 LET A3=A * 180/Y
0780 LET C3=C * 180/Y
0790 GOSUB 0120
0800 LET E=K * (A-P) * P2+R * (A-Z) * Z2
0810 LET F=(1-R) * (C-U) * COS(U) + (1+R) * (2 * Y+P-V) * COS(V)
0820 LET H=R * (Z1-P1) - (1-R) * (C1-U1)
0830 LET J=(1+R) * (V1-A1)
0840 LET L=(A-Z) ↑ 2 - (C-U) ↑ 2 + (2 * Y+P-V) ↑ 2
0850 LET M=E+F+H+J-R * L/G/2
0860 IF W=1 THEN 1060
0870 LET E=A1 * A1 - U1 * U1 + C1 * C1 - V1 * V1
0880 LET F=(P1-Z1) ↑ 2
0890 LET H=(P1-A1) ↑ 2 + (P1-U1) ↑ 2
0900 LET J=(P1-C1) ↑ 2 + (P1-V1) ↑ 2
0910 LET Q=(E-R * (F-H+J))/2
0920 LET LO=M * COS(A8)
0930 LET XO=G * Q * COS(A8)/TAN(A8)
0940 LET AO=XO+YO
0950 LET KO=AO/LO
0960 IF NO=O THEN 0970
0965 RETURN

```

```

0970 OPEN FILE(1,1),"¥LPT"
0980 PRINT FILE(1),"JK";JO,"RN";M9,"AL";A8*180/Y,"LO";LO
0990 PRINT FILE(1),"Y";YO,"AO";AO,"R";R,"P";P*180/Y,"G";G
1000 PRINT FILE(1),"Z";Z*180/Y,"A";A3,"C";C3,"B";B*180/Y
1002 PRINT FILE(1),"D";D*180/Y,"N";N,"W";W,"KE";KO
1010 CLOSE FILE(1)
1020 RETURN
1030 LET W=3
1040 LET C=U
1050 GOTO 0670
1060 LET M=M-(C-Z)*(Z2-K*P2)
1070 GOTO 0870
1080 DEF FNB(X)=H+F*COS(X)+E*X
1090 LET J=0
1100 IF FNB(M)*FNB(Q)>0 THEN 1190
1110 LET J=(M+Q)/2
1120 LET I=FNB(J)
1130 IF ABS(I)<=.00001 THEN 1200
1140 IF FNB(J)*FNB(M)>0 THEN 1170
1150 LET Q=J
1160 GOTO 1110
1170 LET M=J
1180 GOTO 1110
1190 LET J=8
1200 RETURN
1210 LET W=1
1220 LET A=Z
1230 LET E=R/G
1240 LET F=-(1-R)
1250 LET H=K*P2-R*Z2-E*Z
1260 LET M=Z
1270 LET Q=2*Y+P
1280 GOSUB 1080
1290 LET C=J
1300 LET A=Z
1310 LET U=A
1320 GOTO 0660
1330 LET W=2

```

```

1340 LET C = 2 * Y + P
1350 LET V = C
1360 LET L = ((1 - R) * CQS(U) + R * (2 * Y + P - U) / G) / P2 + R
1370 GOTO 0710
1380 LET B = 3 * Y / 2
1390 GOTO 0390
1490 LET A5 = 3.1 / 2
1500 GOTO 0020
1999 END

```

## 参 考 文 献

- [1] 振动输送机的设计计算与使用  
(西德) K. N. 魏曼埃尔  
《Der Maschinenschaben》1963(36)Hef t11~12
- [2] 定向电磁振动运输机振幅的计算  
(西德) K. 杜宇林  
《起重运输机械译丛》1964年第4期
- [3] 物料由于阻尼和质量结合对振动槽运动状态的影响  
(西德) 瓦尔特·霍尔曼  
《起重运输机械》1975年第6期—第7期
- [4] 铸造车间的振动运输机  
(西德) H. G. Rechner  
《国外铸锻机械》1965年第2期
- [5] 振动运输机运动学参数与工艺参数的选择与计算  
《石油化工起重运输简讯》1977年第2期
- [6] 振动运输机参数的研究  
起重运输机械研究所

闻邦椿